Regular Paper

ノルム空間上の Taylor 展開の形式化について

On the Formalization of Taylor Expansion on Normed Spaces

中正 和久 [1,*]
師玉 康成 [2]
Kazuhisa Nakasho[1,*] Yasunari Shidama[2]

1 山口大学大学院創成科学研究科, 山口県宇部市常盤台 2-16-1
1 Graduated School of Science and Technology for Innovation, Yamaguchi University, 2-16-1 Tokiwa-dai, Ube, Japan
2 長野県軽井沢町
2 Karuizawa, Nagano, Japan

* nakasho@yamaguchi-u.ac.jp

Proof checked by Mizar Version: 8.1.14 and MML Version: 5.84.1473

Abstract

In this paper, we report on the formalization of the Taylor expansion on normed spaces.

1 はじめに

筆者らは，[1] で一変数実数関数の Taylor 展開を以下のように形式化した.

Listing 1. TAYLOR_1:33

```
theorem :: TAYLOR_1:33
  for n be Nat, f be PartFunc of REAL,REAL, x0,r be Real st
  ].x0-r,x0+r.[ c= dom f & 0 < r & f is_differentiable_on n+1, ].x0-r,x0+r.[
  for x be Real st x in ].x0-r, x0+r.[
  ex s be Real st 0 < s & s < 1
   & f.x = Partial_Sums(Taylor(f, ].x0-r,x0+r.[,x0,x)).n
        + (diff(f,].x0-r,x0+r.[).(n+1)).(x0+s*(x-x0)) * (x-x0) |^ (n+1) / ((n+1)!);
```

またこれを用いたマクロリン展開などを形式化している [2].

Listing 2. TAYLOR_2:2-14

```
theorem :: TAYLOR_2:2
  for n be Nat, f be PartFunc of REAL,REAL, r be Real st
  0 < r & ].−r,r.[ c= dom f & f is_differentiable_on n+1,].−r,r.[
  for x be Real st x in ].−r, r.[
  ex s be Real st 0 < s & s < 1
  & f.x = Partial_Sums(Maclaurin(f,].−r,r.[,x)).n
        + (diff(f,].−r,r.[).(n+1)).(s*x) * x |^ (n+1) / ((n+1)!);

theorem :: TAYLOR_2:8
  for n be Element of NAT, r,x be Real st 0 < r holds
  Maclaurin(exp_R,].−r,r.[,x).n = x |^ n / (n!);

theorem :: TAYLOR_2:14
  for r, e be Real st 0 < r & 0 < e
  ex n be Nat st
  for m be Nat st n <= m holds
  for x being Real st x in ].−r,r.[
  holds |.exp_R.x−Partial_Sums(Maclaurin(exp_R,].−r,r.[,x)).m.| < e;
```

本稿ではノルム空間上の関数の Taylor 展開に関する形式化を報告する．まず，ノルム空間上の実数値関数の Taylor 展開の形式化を行った．次いで，ノルム空間上のノルム値関数の Taylor 展開の形式化を行った．さらに Taylor 展開を用いてノルム空間上の実数値関数の極値の必要条件及び十分条件の命題とその証明の形式化を行った．

2 高階微分と有界多重線形写像

2 階微分を例にとる．S,T をノルム空間とする．S から T への有界線形写像全ての集合が作るノルム空間を $\mathcal{L}(S,T)$ で表す．これは [3] で

`R_NormSpace_of_BoundedLinearOperators(S,T)`

と定義されている．g を S から T への部分関数（定義域が S 全域とは限らない関数），Z を S の開部分集合とする．g が Z 上で微分可能なとき，g の導関数 $\frac{d}{dx}g$ は $Z \subseteq S$ から $\mathcal{L}(S,T)$ への写像である．

$$g' :\ x \in Z \mapsto g'(x) \in \mathcal{L}(S,T) \tag{1}$$

これは [4] で `g`|Z` と表され，以下のように定義されている．

Listing 3. NDIFF_1:def 9

```
definition
  let S,T;
  let g be PartFunc of S,T;
  let Z be set;
  assume f is_differentiable_on Z;
  func g`|Z −> PartFunc of S, R_NormSpace_of_BoundedLinearOperators(S,T) means
:: NDIFF_1:def 9
  dom it = Z & for x be Point of S st x in X holds it/.x = diff(g,x);
end;
```

さらに g が Z 上で 2 階微分可能なとき，g の 2 階導関数は $Z \subseteq S$ から $\mathcal{L}(S,\mathcal{L}(S,T))$ への写像である．これを一般化して g の Z 上の n 階導関数は [5] で以下のように帰納的に定義されている．以下の記述中 `diff_SP(n,S,T)` は $n=2$ の場合，$\mathcal{L}(S,\mathcal{L}(S,T))$ を表している．

Listing 4. NDIFF_6:def 6

```
definition
  let S,T be RealNormSpace, g be PartFunc of S,T,
      Z be Subset of S, i be Nat;
  func diff(g,n,Z) -> PartFunc of S,diff_SP(n,S,T) equals :: NDIFF_6:def 6
  diff(g,Z).n;
end;

theorem :: NDIFF_6:11
  diff(g,Z).0 = g|Z & diff(g,Z).1 = (g|Z)`| Z & diff(g,Z).2 = ((g|Z)`| Z) `| Z;

theorem :: NDIFF_6:13
  diff(g,n+1,Z) = diff(g,n,Z) `|Z;
```

ここで g|Z は関数 g の定義域を Z に制限した関数で，(g|Z)`|Z は上記の NDIFF_1:def 9 による g|Z の Z 上の導関数を表す．

g が Z 上で n 階微分可能であることを表す述語 g is_differentiable_on n,Z も [5] で以下のように帰納的に定義されている．

Listing 5. NDIFF_6:def 7

```
definition
  let S,T be RealNormSpace;
  let g be PartFunc of S,T;
  let Z be Subset of S;
  let n be Nat;
  pred g is_differentiable_on n,Z means :: NDIFF_6:def 7
  Z c= dom g & for i be Nat st i <= n-1 holds
  modetrans(diff(g,Z).i,S,diff_SP(i,S,T)) is_differentiable_on Z;
end;

theorem :: NDIFF_6:14
  g is_differentiable_on n,Z
  iff Z c= dom g & for i be Nat st i <= n-1 holds diff(g,i,Z) is_differentiable_on Z;

theorem :: NDIFF_6:15
  g is_differentiable_on 1,Z
  iff Z c= dom g & g|Z is_differentiable_on Z;

theorem :: NDIFF_6:16
  g is_differentiable_on 2,Z
  iff Z c= dom g & g|Z is_differentiable_on Z & (g|Z)`|Z is_differentiable_on Z;

theorem :: NDIFF_6:17
  for S,T be RealNormSpace, g be PartFunc of S,T, Z be Subset of S, n be Nat
  st g is_differentiable_on n,Z
  for m be Nat st m <= n holds g is_differentiable_on m,Z;
```

diff(g,n,Z) は S から T への有界線形写像が作るノルム空間 $\mathcal{L}(S,T)$ を n 回使って作られるノルム空間の入れ子

$$
\begin{gathered}
\mathcal{L}(S,T) \\
\mathcal{L}(S,\mathcal{L}(S,T)) \\
\mathcal{L}(S,\mathcal{L}(S,\mathcal{L}(S,T))) \\
\vdots \\
\mathcal{L}(S,\mathcal{L}(\mathcal{L}\cdots\mathcal{L}(S,T)\cdots)) \qquad (2)
\end{gathered}
$$

に値をとる S 上の関数である．これは有界な n 重線形写像であり，有界な n 重線形写像

全てが作るノルム空間

$$\mathcal{M}(S^n, T)$$

の元とみなすことができる．この同値な読み替えをする作用素が以下 LOPBAN14:def 4 で定義されている．これは有界線形写像を元とするノルム空間の多重の入れ子と有界な多重線形写像を元とするノルム空間の間のノルム等長な同型写像である．

Listing 6. LOPBAN14:def 4

```
definition
  let Y be RealNormSpace, X be RealNormSpace-Sequence;
  func NestMult(X,Y) -> Lipschitzian LinearOperator of
        NestingLB(X,Y),R_NormSpace_of_BoundedMultilinearOperators(X,Y)
  means :: LOPBAN14:def 4
  it is one-to-one onto isometric
    & (for u being Element of NestingLB(X,Y) holds ||.it.u.|| = ||.u.||)
    & for u be Point of NestingLB(X,Y), x be Point of product X
        ex g be     FinSequence
        st len g = len X & g.1 = u
         & (for i be Element of NAT st 1 <= i & i < len X holds
            ex Xi be RealNormSpace-Sequence, h be Point of NestingLB(Xi,Y)
            st Xi = X | (len X -' i + 1) & h = g.i & g.(i + 1) = h.(x.(len X -' i + 1)))
    & ex X1 be RealNormSpace-Sequence, h be Point of NestingLB(X1,Y)
      st X1 = <*X.1*> & h = g.(len X) & (it.u).x = h.(x.1);
end;
```

たとえば $n=2$ の場合，X を長さ 2 のノルム空間の有限列 $X=[X_1,X_2]$，T をノルム空間とするとき X,Y による入れ子

$$\mathcal{L}(X_2, \mathcal{L}(X_1, T))$$

をこの形式化では `NestingLB(X,T)` で表している．

この入れ子の元

$$u \in \mathrm{NestingLB}(X, T) = \mathcal{L}(X_2, \mathcal{L}(X_1, T))$$

に上記の同型写像 `NestMult(X,T)` はノルム空間 X_1, X_2 のそれぞれの任意の元 $h_1 \in X_1, h_2 \in X_2$ に対して

$$v(h_1, h_2) = u(h_1)(h_2)$$

となる $X_1 \times X_2$ から Y への 2 重線形写像 v を対応させる．

以下の TAYLOR_3:def 1 で定義される `DIFF(g,n,Z)` は，`NestMult(n,S,T)` によって作られる `diff(g,n,Z)` と同値な S の多重直積 S^n から T への有界な n 重線形写像が作るノルム空間

$$\mathcal{M}(S^n, T) \tag{3}$$

に値をとる S 上の関数である．

Listing 7. TAYLOR_3:def 1

```
definition
  let S, T be RealNormSpace;
  let g be PartFunc of S,T;
  let Z be Subset of S;
  let n be non empty Nat;
  func DIFF (g,n,Z) -> PartFunc of S,R_NormSpace_of_BoundedMultilinearOperators(n|->S,T)
  equals :: TAYLOR_3:def 1
  NestMult(n,S,T) * diff(g,n,Z);
end;
```

上記で

`R_NormSpace_of_BoundedMultilinearOperators(n|-> S,T)`

が $\mathcal{M}(S^n, T)$ を表している．`n|->S` はノルム空間 S を n 個ならべた長さ n の有限列

$$\langle S, S, \cdots, S \rangle$$

を表している．また，`product(n|->S)` はノルム空間 S の n 重直積空間 S^n を表している．ノルム空間の有限列

$$G = \langle G_1, G_2, \cdots, G_n \rangle$$

が与えられたとき `product G` は多重直積空間 $\prod_{i=1}^{i=n} G_i$ を表している. `product G` のベクトルは各 G_i の元 $x_i \in G_i$ からなる有限列 $x = \langle x_1, x_2, \cdots, x_n \rangle$ であり，`product G` のベクトル

$$x = \langle x_1, x_2, \cdots, x_n \rangle,\ y = \langle y_1, y_2, \cdots, y_n \rangle\ \in \prod_{i=1}^{i=n} G_i \tag{4}$$

について加法は

$$x + y = \langle x_1 + y_1, x_2 + y_2, \cdots, x_n + y_n \rangle\ \in \prod_{i=1}^{i=n} G_i \tag{5}$$

で定義され, 実数 r との積は

$$r \cdot x = \langle r \cdot x_1, r \cdot x_2, \cdots, r \cdot x_n \rangle\ \in \prod_{i=1}^{i=n} G_i \tag{6}$$

で定義される．また x のノルム $\|x\|$ は

$$\|x\| = \sqrt{\|x_1\|^2 + \|x_2\|^2 + \cdots + \|x_n\|^2} \tag{7}$$

で定義される. これらは [6] で形式化されている.

3　各点での高階微分

g の S のベクトル x での高階微分可能性とその微分値を以下のように定義する.

g を S から T への部分関数とする．g が S のベクトル x で $k(> 0)$ 階微分可能であるとは，x の開近傍 Z が存在し，g が Z 上で $k-1$ 階微分可能であり，かつその $k-1$ 階導関数 $g^{(k-1)}$ が x で微分可能であることと定義する．以下がその形式化である.

Listing 8. TAYLOR_4:def 1

```
definition
  let S,T be RealNormSpace, g be PartFunc of S,T,
      x be Point of S, k be Nat;
  pred g is_differentiable_in k,x
  means :: TAYLOR_4:def 1
  ex Z be Subset of S
  st Z is open & x in Z & g is_differentiable_on k-'1,Z
   & diff(g,k-'1,Z) is_differentiable_in x;
end;
```

さらに，g の x での k 階微分値を g の $k-1$ 階導関数の x の微分値

$$\left(g^{(k-1)}\right)'(x)$$

で定義する．以下がその形式化である．

Listing 9. TAYLOR_4:def 2

```
definition
  let S,T be RealNormSpace, g be PartFunc of S,T,
      x be Point of S, k be Nat;
  assume 1 <= k & g is_differentiable_in k,x;
  func diff(g,k,x) -> Point of diff_SP(k,S,T)
  means :: TAYLOR_4:def 2
  ex Z be Subset of S
  st Z is open & x in Z & g is_differentiable_on k-'1,Z
   & diff(g,k-'1,Z) is_differentiable_in x
   & it = diff(diff(g,k-'1,Z),x);
end;
```

以上の定義が正当性をもつためには，これらの定義が x の開近傍 Z の選択に依存しないことを示す必要があり，そのための補題群を形式化した．以下にその一部を掲載する．

Listing 10. TAYLOR_4:1-4

```
theorem :: TAYLOR_4:1
  for S,T be RealNormSpace, f be PartFunc of S,T, Z,W be Subset of S
  st f is_differentiable_on Z & f is_differentiable_on W
  holds f is_differentiable_on Z \/ W & f is_differentiable_on Z /\ W;

theorem :: TAYLOR_4:2
  for S,T be RealNormSpace, f be PartFunc of S,T,
      Z,W be Subset of S, x be Point of S
  st f is_differentiable_on Z & f is_differentiable_on W & x in Z & x in W
  holds (f `| Z)/.x = (f `| W)/.x;

theorem :: TAYLOR_4:4
  for S,T be RealNormSpace, g be PartFunc of S,T,
      Z,W be Subset of S, n be Nat
  st g is_differentiable_on n,Z & g is_differentiable_on n,W
  holds g is_differentiable_on n,Z \/ W
  & (for i be Nat st i <= n holds diff(g,i,Z \/ W) | Z = diff(g,i,Z)
     & diff(g,i,Z \/ W) | W = diff (g,i,W ))
  & g is_differentiable_on n,Z /\ W
  & (for i be Nat st i <= n holds diff(g,i,Z /\ W) = diff(g,i,Z) | (Z /\ W)
     & diff(g,i,Z /\ W) = diff(g,i,W ) | (Z /\ W));
```

以上の定義について，点における高階微分の線形性や開集合上での高階微分可能性の定義との整合性などを示す命題群を形式化した．以下にその一部を掲載する．

Listing 11. TAYLOR_4:1-5

```
theorem :: TAYLOR_4:15
  for S,T be RealNormSpace, g,h be PartFunc of S,T,
      x be Point of S, n be Nat
  st 1 <= n & g is_differentiable_in n,x & h is_differentiable_in n,x
  holds (g + h) is_differentiable_in n,x & diff(g+h,n,x) = diff(g,n,x) + diff(h,n,x);

theorem :: TAYLOR_4:16
  for S,T be RealNormSpace, g be PartFunc of S,T,
      x be Point of S, r be Real, n be Nat
  st 1 <= n & g is_differentiable_in n,x
  holds (r(#)g) is_differentiable_in n,x & diff(r(#)g,n,x) = r * diff(g,n,x);
```

```
theorem :: TAYLOR_4:19
  for S,T be RealNormSpace, g be PartFunc of S,T,
      Z be Subset of S, k be Nat
  st 1 <= k & Z is open & Z c= dom g
  holds
  g is_differentiable_on k,Z
  iff for x be Point of S st x in Z holds g is_differentiable_in k,x;
```

さらに以下のように，前章で述べたノルム空間の有界線形写像が作る入れ子空間と，有界多重写像が作る空間の間に，同型による微分値の同値変換を導入した．

Listing 12. TAYLOR_4:def 3

```
definition
  let S,T be RealNormSpace;
  let f be PartFunc of S,T;
  let x be Point of S;
  let n be non empty Nat;
  func DIFF (f,n,x)
    -> Point of R_NormSpace_of_BoundedMultilinearOperators(n|->S,T)
  equals :: TAYLOR_4:def 3
  NestMult(n,S,T) . diff(f,n,x);
end;
```

4 実数値関数の Taylor 展開

4.1 命題の形式化

筆者らは次の命題とその証明を形式化した [7].

S を自明でないノルム空間，g を定義域が S の全域とは限らない S 上の実数値関数（部分関数），Z を g の定義域に含まれる S の部分集合，g は Z 上で n 階微分可能とする．x, h を S 上のベクトルで，h は零元ではないものとする．また，区間

$$[x, x+h] = \{x + t \cdot h \mid 0 \le t \le 1\}$$

が Z に含まれ，g の Z 上の n 階導関数 $g^{(n)}$ が $[x, x+h]$ 上で連続，かつ g は $(x, x+h)$ に含まれる任意の s で $n+1$ 階微分可能とする．このとき実数 $c \in (0,1)$ が存在して，

$$g(x+h) = \sum_{k=0}^{n} \frac{g^{(k)}(x)}{k!} \cdot h^k + \frac{g^{(n+1)}(x + c \cdot h)}{(n+1)!} \cdot h^{n+1} \tag{8}$$

が成り立つ．この命題の形式化が以下である．

Listing 13. TAYLOR_4:23

```
theorem :: TAYLOR_4:23
  for S be non trivial RealNormSpace,
      g be PartFunc of S,#REAL, Z be Subset of S, n be non empty Nat
  st Z c= dom g & g is_differentiable_on n,Z
  holds
    for x,h be Point of S
    st h <> 0.S & [.x,x+h.] c=Z & diff(g,n,Z) is_continuous_on [.x,x+h.]
     & for s be Point of S st s in ].x,x+h.[
        holds g is_differentiable_in n+1,s
    holds
      ex c be Real st c in ].0,1.[
       & g.(x+h) = Partial_Sums(Taylor(g,Z,x,x+h)).n
          + 1/((n + 1)!) * (DIFF(g,n+1,x+c*h)) . (h|^(n+1));
```

上記の記述中の`#REAL`は，実数の集合を加法，乗算，0 に関するノルム空間とみなしたものを表し，以下のように定義されている．

Listing 14. LOPBAN15:def 2

```
definition
  func #REAL -> non empty NORMSTR equals :: LOPBAN15:def 2
  NORMSTR(# REAL, In(0,REAL), addreal, multreal, absnorm #);
end;
registration
  cluster #REAL -> non empty right_complementable Abelian
    add-associative right_zeroed vector-distributive
    scalar-distributive scalar-associative scalar-unital
    discerning reflexive RealNormSpace-like;
end;
```

ノルム空間`#REAL`を用いる理由は，筆者らが形式化して MML に収録されている高階微分を含む一連の微分の記述が全てノルム空間上の関数を扱っており，それらの命題群を活用するためである．第 2 章で述べたように`diff(g,n,Z)`は g の Z 上の n 階導関数を表している．`Taylor(g,Z,x,x+h)`は関数 g の Taylor 級数列を表し，以下のように定義した．

Listing 15. TAYLOR_3:def 4

```
definition
  let S,T be RealNormSpace, g be PartFunc of S,
      T,Z be Subset of S, a,b be Point of S;
  func Taylor(g,Z,a,b) -> sequence of T means :: TAYLOR_3:def 4
    it.0 = (g|Z)/.a
   & for n be non empty Nat holds
    it.n = 1/(n!) * ((DIFF(g,n,Z)/.a).((b-a)|^n));
end;
```

`(b-a)|^n`は`#REAL`の元`b-a`を n 個ならべた長さ n の有限列を表しており，以下のように定義している．

Listing 16. TAYLOR_3:def 2

```
definition
  let S be RealNormSpace,h be Point of S, n be non empty Nat;
  func h |^ n -> Point of product(n|->S) equals :: TAYLOR_3:def 2
  n |-> h;
end;
```

`Partial_Sums`はノルム空間など加法が定義されている空間の元の列の総和で，以下で帰納的に定義されている [8].

Listing 17. BHSP_4:def 1

```
definition
  let X be non empty addLoopStr;
  let seq be sequence of X;
  func Partial_Sums(seq) -> sequence of X means :: BHSP_4:def 1
  it.0 = seq.0 & for n holds it.(n + 1) = it.n + seq.(n + 1);
end;
```

`DIFF(g,n,Z)`も第 2 章で述べたように`diff(g,n,Z)`と同様 g の Z 上の n 階導関数を表しているが，S の元 a について`DIFF(g,n,Z)/.a`は Z から S の n 重直積 S^n から T への有界な多重線形写像である．

4.2 命題の証明

命題 TAYLOR_4:23 の証明は，以下の補題を用いて前章で述べた TAYLOR_1:33 を適用する．

g をノルム空間 S から $\mathbb{R}$ への部分関数，Z を S の（開）部分集合で，g が Z 上で $n+1$ 階微分可能とする．S のベクトル x_0 と $r \neq 0_S$ について集合

$$(x_0 - r, x_0 + r) = \{x_0 + t \cdot r \mid -1 < t < 1\}$$

が Z に含まれるものとする．このとき, 実数区間 (-1,1) 上の実数値関数

$$f: \ t \in (-1,1) \mapsto g(x_0 + t \cdot r) \in \mathbb{R} \tag{9}$$

が存在し，f は $(-1,1)$ 上で $n+1$ 階微分可能であり，かつ $(-1,1)$ に含まれる任意の実数 $t \in (-1,1)$ について

$$f^{(n+1)}(t) = g^{(n+1)}(x_0 + t \cdot r) \cdot r^{n+1} \tag{10}$$

これを形式化したものが以下である．

Listing 18. TAYLOR_3:50

```
theorem :: TAYLOR_3:50
  for S be non trivial RealNormSpace, F be Function of REAL,REAL,
      g be PartFunc of S,#REAL, Z be Subset of S, x0,r be Point of S
  st Z c= dom g & ].x0 − r,x0 + r.[ c= Z & r <> 0.S
  holds
  ex f be PartFunc of REAL,REAL
  st dom f = ].−1,1.[
   & ( for t being Real st t in ].−1,1.[ holds f.t = g.(x0 + t*r) )
   & for n be Nat st g is_differentiable_on n+1,Z
     holds f is_differentiable_on n+1,].−1,1.[
   & for t be Real st t in ].−1,1.[ holds
     (diff(f,].−1,1.[).(n+1)).t = (DIFF(g,n+1,Z)/.(x0 + t*r)).((n+1)|−>r);
```

5 ノルム空間に値をもつ関数の Taylor 展開

5.1 命題の形式化

筆者らは次の命題とその証明を形式化した [7].

S をノルム空間，g を定義域が S の全域とは限らない S 上の実数値関数（部分関数）, Z を g の定義域に含まれる S の部分集合，g は Z 上で n 階微分可能とする．x, h を S のベクトルで，h は S の零元ではないものとする．M は 0 以上の実数とする．区間

$$[x, x+h] = \{x + t \cdot h \mid 0 \leq t \leq 1\}$$

が Z に含まれ，g の Z 上の n 階導関数 $g^{(n)}$ が $[x, x+h]$ 上で連続，かつ g は $(x, x+h)$ 上で $n+1$ 階微分可能で

$$\left\| g^{(n+1)}(s) \right\| \leq M$$

を満たすとする．このとき，

$$\left\| g(x+h) - \sum_{k=0}^{n} \frac{g^{(k)}(x)}{k!} \cdot h^k \right\| \leq \frac{M}{(n+1)!} \cdot \|h\|^{n+1} \tag{11}$$

以下がその形式化である．

Listing 19. TAYLOR_5:49

```
theorem :: TAYLOR_5:49
for S,T be RealNormSpace, Z be non empty open Subset of S,
    x,h be Point of S
st [.x,x+h.] c= Z holds
  for g be PartFunc of S,T, M be Real, n be Nat
  st 0 <= M & g is_differentiable_on n,Z
   & ( for s be Point of S st s in [.x,x+h.]
       holds diff(g,n,Z) is_continuous_in s )
   & ( for s be Point of S st s in ].x,x+h.[
       holds g is_differentiable_in n+1,s
           & ||.diff (g,n+1,s).|| <= M )
  holds
    ||.g/.(x+h) - Partial_Sums(Taylor(g,Z,x,x+h)).n.|| <= M/((n+1)!) * ||.h.|| |^ (n+1);
```

5.2 命題の証明

微分の階数についての帰納法を用いる [7]．$n = 0$ については，この命題は有限増分の公式であり，[9] で以下のように形式化されている．

Listing 20. NDFIF_5:19

```
theorem :: NDIFF_5:19
  for S,T be RealNormSpace,
      f be PartFunc of S,T, p,q be Point of S, M be Real
   st [.p,q.] c= dom f
    & (for x be Point of S st x in [.p,q.] holds f is_continuous_in x)
    & (for x be Point of S st x in ].p,q.[ holds f is_differentiable_in x)
    & (for x be Point of S st x in ].p,q.[ holds ||. diff(f,x) .|| <= M)
  holds ||. f/.q - f/.p .|| <= M * ||. q-p .||;
```

n について命題が成り立っているものと仮定する．g は Z 上 $n+1$ 階微分可能で，g の $n+1$ 階導関数 $g^{(n+1)}$ が $[x, x+h]$ 上で連続，かつ g は $(x, x+h)$ 上で $n+2$ 階微分可能であるとする．

ここで実数区間 $[0, 1]$ から T への写像

$$f : t \in [0,1] \mapsto g(x + t \cdot h) - \sum_{k=0}^{n+1} \frac{g^{(k)}(x)}{k!} \cdot (t \cdot h)^k \ \in T \tag{12}$$

を導入すると，$t \in [0, 1]$ について

$$f'(t) = \left(g'(x + t \cdot h) - \sum_{k=0}^{n} \frac{(g')^{(k)}(x)}{k!} \cdot (t \cdot h)^k \right) \cdot h \tag{13}$$

ここで g についての仮定から，g' は $Z \subseteq S$ 上から $\mathcal{L}(S, T)$ への写像であり，Z 上 n 階微分可能で g' の n 階導関数 $(g')^{(n)}$ は $[x, x + t \cdot h]$ 上で連続，かつ g' は $(x, x + t \cdot h)$ 上で $n + 1$ 階微分可能である．

g' と増分 $t \cdot h$ について帰納法の仮定から，命題の評価式が成り立つので，

$$\left\| g'(x + t \cdot h) - \sum_{k=0}^{n} \frac{(g')^{(n)}(x)}{k!} \cdot (t \cdot h)^k \right\| \leq \frac{M}{(n+1)!} \cdot \|t \cdot h\|^{n+1} \tag{14}$$

これと (13) 式から

$$\|f'(t)\| \leq \frac{M}{(n+1)!} \cdot \|t \cdot h\|^{n+1} \cdot \|h\| \tag{15}$$

となり $t \in [0, 1]$ から

$$\|f'(t)\| \leq \frac{M}{(n+1)!} \cdot \|h\|^{(n+1)+1} \cdot t^{n+1} \tag{16}$$

区間 $[0, 1]$ 上の実数値関数 u を

$$u: \ t \in [0, 1] \mapsto \frac{M}{((n+1)+1)!} \cdot \|h\|^{(n+1)+1} \cdot t^{(n+1)+1}$$

と定義すると，

$$u'(t) = \frac{M}{(n+1)!} \cdot \|h\|^{(n+1)+1} \cdot t^{n+1}$$

したがって (16) 式から

$$\|f'(t)\| \leq u'(t) \tag{17}$$

上式に補題 TAYLOR_5:22

Listing 21. TAYLOR_5:22

```
theorem :: TAYLOR_5:22
  for T be RealNormSpace,
      f be PartFunc of REAL,T,
      u be PartFunc of REAL,REAL
  st dom f = [.0,1.] & dom u = [.0,1.]
   & f | [.0,1.] is continuous
   & u | [.0,1.] is continuous
   & f is_differentiable_on ].0,1.[
   & u is_differentiable_on ].0,1.[
   & (for x be Real st x in ].0,1.[
      holds ||. diff(f,x) .|| <= diff(u,x))
holds
  ||. f/.1 - f/.0 .|| <= u/.1 - u/.0;
```

を適用すると

$$\|f(1) - f(0)\| \leq u(1) - u(0) = \frac{M}{((n+1)+1)!} \cdot \|h\|^{(n+1)+1} \tag{18}$$

f の定義 (12) 式から

$$\|f(1) - f(0)\| = \left\| g(x + h) - \sum_{k=0}^{n+1} \frac{g^{(k)}(x)}{k!} \cdot h^k \right\| \tag{19}$$

であり，(18) 式から帰納法が完成する．

6 関数の極値条件

この章では関数の極値についての必要条件と十分条件の形式化について述べる．以下，極小値についてのみ記述する.（極大値については不等号の向きが逆になるだけである.）

6.1 関数の極値の定義

準備として，実数空間上の実数値関数，一般のノルム空間上の実数値関数の極小・極大値，真の極小・極大値の定義を形式化した．何れも，極小・極大になる点の近傍上での関数値の大小関係である.

記述中の `In(f.x,REAL)`,`In(f.y,REAL)` は実数集合をノルム空間とみなした `#REAL` のベクトル `f.x`,`f.y` に，実数集合 `REAL` の元として実数の四則演算や大小比較を可能にするための作用を施していることを表す.

Listing 22. TAYLOR_6:def 1-11

```
definition
  let f be PartFunc of REAL,REAL;
  let x be Real ;
  pred f is_local_minimum_point_at x
  means :: TAYLOR_6:def 1
  ex Z be open Subset of REAL
  st Z c= dom f & x in Z & for y be Real st y in Z holds f.x <= f.y;
end;

definition
  let f be PartFunc of REAL,REAL;
  let x be Real;
  pred f is_local_trueminimum_point_at x
  means :: TAYLOR_6:def 3
  ex Z be open Subset of REAL
  st Z c= dom f & x in Z
   & for y be Real st y in Z & y <> x holds f.x < f.y;
end;

definition
  let S be RealNormSpace;
  let f be PartFunc of S,#REAL;
  let x be Point of S ;
  pred f is_local_minimum_point_at x
  means :: TAYLOR_6:def 9
  ex Z be open Subset of S
  st Z c= dom f & x in Z
   & for y be Point of S st y in Z holds In(f.x,REAL) <= In(f.y,REAL);
end;

definition
  let S be RealNormSpace;
  let f be PartFunc of S,#REAL;
  let x be Point of S;
  pred f is_local_trueminimum_point_at x
  means :: TAYLOR_6:def 11
  ex Z be open Subset of S
  st Z c= dom f & x in Z
   & for y be Point of S st y in Z & y <> x holds In(f.x,REAL) < In(f.y,REAL);
end;
```

6.2 必要条件，十分条件

実数空間上の実数値関数に関する必要条件の形式化は，一変数関数の微分を用いた極値の条件（停留条件）による．

Listing 23. TAYLOR_6:1

```
theorem :: TAYLOR_6:1
  for f be PartFunc of REAL,REAL, x be Real
  st f is_local_minimum_point_at x
   & f is_differentiable_in x holds diff(f,x) = 0;
```

$\mathbb{R}^n$ 空間上の実数値関数，一般のノルム空間上の実数値関数 g ついては，極値をとるベクトル x において，零ベクトルでない h を用いた x の周りの h 摂動を表す実数空間上の実数値関数

$$f:\ t \in (-1, 1) \mapsto g(x + t \cdot h)$$

が $t = 0$ で極値をとるための条件に帰着させる．そのための合成関数 f の $t = 0$ における微分値が

$$f'(0) = g'(x) \cdot h$$

であることを示す補題は以下のように形式化される．

Listing 24. TAYLOR_6:3

```
theorem :: TAYLOR_6:3
  for S be RealNormSpace, g be PartFunc of S,#REAL,
      f be PartFunc of REAL,REAL, W0 be open Subset of REAL,
      x,h be Point of S
  st g is_differentiable_in x
   & 0 < ||.h.|| & 0 in W0 & dom f = W0
   & ( for t be Real st t in W0 holds x+t*h in dom g )
   & for t be Real st t in W0 holds f.t = g.(x+t*h)
     holds f is_differentiable_in 0 & diff(f,0)= diff(g,x).h;
```

以下，極小値の必要条件についての形式化の一部を抜粋する．

`product G` はノルム空間の有限列 $\langle G_1, G_2, \ldots, G_n \rangle$ による多重直積空間 $\prod_{i=1}^{n} G_i$ を表し，`partdiff(f,x,i)` はノルム空間 $\prod_{i=1}^{n} G_i$ から`#REAL` または `REAL` への写像 f の偏微分

$$\frac{\partial f}{\partial x_i}(x)$$

を表している．

Listing 25. TAYLOR_6:4-6

```
theorem :: TAYLOR_6:4
  for S be RealNormSpace, g be PartFunc of S,#REAL, x be Point of S
  st g is_differentiable_in x & g is_local_minimum_point_at x
  holds diff(g,x) = 0.R_NormSpace_of_BoundedLinearOperators(S,#REAL);

theorem :: TAYLOR_6:6
  for G being RealNormSpace-Sequence, f be PartFunc of product G,#REAL,
      x be Point of product G
  st f is_differentiable_in x & f is_local_minimum_point_at x
  holds
    for i be Element of dom G holds
    partdiff(f,x,i) = 0.R_NormSpace_of_BoundedLinearOperators(G.i,#REAL);
```

さらに前章の Taylor 展開を用いて以下の命題を得た．

Listing 26. TAYLOR_7:2

```
theorem :: TAYLOR_7:2
  for S be non trivial RealNormSpace, g be PartFunc of S,#REAL,
      Z be open Subset of S, x be Point of S, n be non empty Nat
  st x in Z & Z c= dom g & g is_differentiable_on n+1,Z
    & ( for m be non empty Nat st m < n holds
        for r be Point of S holds (DIFF (g,m,Z)/.x).(r|^m) = 0 )
    & ( ex K be Real st 0 < K &
        for x,r be Point of S st x in Z
        holds ||. (DIFF (g,n+1,Z)/.x).(r|^(n+1)) .|| <= K )
    & ( ex h be Point of S st h <> 0.S & (DIFF(g,n,Z)/.x).(h |^ n) <> 0. #REAL )
    & g is_local_maximum_point_at x
  holds ( for h be Point of S holds In((DIFF(g,n,Z)/.x).(h|^n),REAL) <= 0 )
      & ( n is even );
```

さらに十分条件については以下の命題を得た．

Listing 27. TAYLOR_7:5

```
theorem :: TAYLOR_7:5
  for S be non trivial RealNormSpace, g be PartFunc of S,#REAL,
      Z be non empty open Subset of S, x be Point of S,
      M,delta be Real, n be non empty Nat
   st 0 < delta & 0 < M & x in Z & Z c= dom g
    & g is_differentiable_on n+1,Z
    & ( for m be non empty Nat st m < n holds
        for r be Point of S holds (DIFF(g,m,Z)/.x).(r|^m)= 0. #REAL )
    & ( for s be Point of S st s in Z holds ||. diff(g,n+1,Z)/.s .|| <= M )
    & for h be Point of S holds
      delta * (||.h.||) |^ n <= In((DIFF(g,n,Z)/.x).(h|^n),REAL)
  holds
    g is_local_trueminimum_point_at x;
```

7 まとめ

本稿ではノルム空間上の Taylor 展開定理 [7] と証明の形式化を報告した．筆者らは，この他に変分法の定理と証明を形式化した．さらに，それらの諸定理の準備の上に微分多様体の形式化を開始している．これらについては次回以降に報告する．

参考文献

[1] Shidama Y. The Taylor Expansions. Formalized Mathematics. 2004;12(**2**):195–200. Available from: `http://fm.mizar.org/2004-12/pdf12-2/taylor_1.pdf`.

[2] Nishino A, Shidama Y. The Maclaurin Expansions. Formalized Mathematics. 2005;13(**3**):421–425. Available from: `http://fm.mizar.org/2005-13/pdf13-3/taylor_2.pdf`.

[3] Shidama Y. Banach Space of Bounded Linear Operators. Formalized Mathematics. 2004;12(**1**):39–48. Available from: `http://fm.mizar.org/2004-12/pdf12-1/lopban_1.pdf`.

[4] Imura H, Kimura M, Shidama Y. The Differentiable Functions on Normed Linear Spaces. Formalized Mathematics. 2004;12(**3**):321–327. Available from: http://fm.mizar.org/2004-12/pdf12-3/ndiff_1.pdf.

[5] Endou N, Shidama Y. Differentiation in Normed Spaces. Formalized Mathematics. 2013;21(**2**):95–102.

[6] Endou N, Shidama Y, Miyajima K. The Product Space of Real Normed Spaces and its Properties. Formalized Mathematics. 2007;15(**3**):81–85.

[7] Schwartz L. Cours d'analyse. Hermann; 1981.

[8] Kraszewska Ez, Popiołek J. Series in Banach and Hilbert Spaces. Formalized Mathematics. 1991;2(**5**):695–699. Available from: http://fm.mizar.org/1991-2/pdf2-5/bhsp_4.pdf.

[9] Shidama Y. Differentiable Functions on Normed Linear Spaces. Formalized Mathematics. 2012;20(**1**):31–40.

Mizar article information

Works in Progress

NDIFF12 *Derivatives of Lipschitzian Bilinear Operators*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized some fundamentally important theorems for higher-order derivatives of bounded bilinear operators on normed spaces.

NDIFF13 *Differentiation of Composite Functions*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized some fundamentally important theorems for higher-order derivatives of composite functions on norm spaces. Using these theorems, we proved the inverse function theorem for norm-valued functions of normed spaces.

NDIFF14 *Higher Order Partial Derivatives of Functions on Normed Spaces*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the higher-order partial derivatives of functions on normed spaces.

NDIFF15 *Higher Order Derivatives of Functions and Continuous MultilinearFunctions*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the representations of higher-order derivatives of functions on normed spaces as continuous multilinear functions.

TAYLOR_3 *Taylor Expansion of Real − Valued Functions on Normed Spaces − PartI*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the Taylor expansion of real-valued functions on Normed Spaces.

TAYLOR_4 *Taylor Expansion of Real − Valued Functions on Normed Spaces − PartII*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the Taylor expansion of real-valued functions on normed spaces.

TAYLOR_5 $Taylor\ Expansion\ of\ Normed\ Space{-}Valued\ Functions\ on\ Normed\ Spaces$
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the Taylor expansion of normed-space-Valued functions on normed spaces.

TAYLOR_6 $On\ the\ Necessary\ Condition\ for\ Extrema\ of\ Functions\ on\ Normed\ Spaces$
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the necessary condition for extrema of functions on normed spaces.

TAYLOR_7 $On\ the\ Sufficient\ Condition\ for\ Extrema\ of\ Functions\ on\ Normed\ Spaces$
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized the sufficient condition for extrema of functions on normed spaces.